// File: contracts/Vault.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity ^0.8.9;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC20/ERC20.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@chainlink/contracts/src/v0.8/interfaces/AggregatorV3Interface.sol";
import "./TransferHelper.sol";
import "./Formula.sol";
import "./markets/SavingsDaiMarket.sol";
import "./markets/StEthMarket.sol";
import "./exchange/ICindexSwap.sol";

contract Vault is ERC20, Ownable, ReentrancyGuard {

    struct AssetOracle {
        address asset;
        address oracle;
    }

    uint256 public protocolFee = 20;

    // Support staking coins (usdc,usdt,dai)
    address[] public supportAssets;

    mapping (address => AggregatorV3Interface) public oracles;

    mapping (address => uint256) assetAmounts;

    uint256 internal constant PRECISION = 10 ** 18;

    // sDai and stETH usd value weight
    uint256[] weights = [95, 5];

    address private immutable PROTOCOL_FEE_RESERVE;

    // cindex swap router
    ICindexSwap public router;
    // Maker: Dai Stablecoin
    address constant DAI = 0x6B175474E89094C44Da98b954EedeAC495271d0F;
    // Lido: stETH Token
    address constant STETH = 0xae7ab96520DE3A18E5e111B5EaAb095312D7fE84;

    // MakerDAO: sDAI Token, Lido: stETH Token
    address[] public underlyingTokens = [0x83F20F44975D03b1b09e64809B757c47f942BEeA, 0xae7ab96520DE3A18E5e111B5EaAb095312D7fE84];

    event Deposit(address indexed user, uint256 share, address asset, uint256 amount, uint256 amount0, uint256 amount1, string referralCode);

    event DepositUnderlying(address indexed user, uint256 share, uint256 amount0, uint256 amount1);

    event Withdraw(address indexed user, uint256 share, address asset, uint256 amount, uint256 amount0, uint256 amount1);

    event WithdrawUnderlying(address indexed user, uint256 share, uint256 amount0, uint256 amount1);

    event ProtocolFee(uint256 amount0, uint256 amount1);

    struct DepositParams {
        address tokenIn;
        uint256 amountIn;
        ICindexSwap.SwapData[] swapData;
        string referralCode;
    }

    struct WithdrawParams {
        uint256 share;
        address tokenOut;
        ICindexSwap.SwapData[] swapData;
    }

    constructor(
    ) payable
        ERC20("US Treasury Bond + STETH", "TBE")
    {
        supportAssets = [address(0xA0b86991c6218b36c1d19D4a2e9Eb0cE3606eB48),address(0xdAC17F958D2ee523a2206206994597C13D831ec7),address(0x6B175474E89094C44Da98b954EedeAC495271d0F)];
        // address(0x7c4FC00b404775E9Bf3996b1ad0c2b72799e994d) is a multisig wallet.
        transferOwnership(address(0x7c4FC00b404775E9Bf3996b1ad0c2b72799e994d));
        PROTOCOL_FEE_RESERVE = address(0x7c4FC00b404775E9Bf3996b1ad0c2b72799e994d);
        router = ICindexSwap(address(0x29dbE32E73282EDDF08E191b2E7D34bCC79665Dd));
        oracles[address(0xA0b86991c6218b36c1d19D4a2e9Eb0cE3606eB48)] = AggregatorV3Interface(address(0x986b5E1e1755e3C2440e960477f25201B0a8bbD4));
        oracles[address(0xdAC17F958D2ee523a2206206994597C13D831ec7)] = AggregatorV3Interface(address(0xEe9F2375b4bdF6387aa8265dD4FB8F16512A1d46));
        oracles[address(0x6B175474E89094C44Da98b954EedeAC495271d0F)] = AggregatorV3Interface(address(0xAed0c38402a5d19df6E4c03F4E2DceD6e29c1ee9));
        oracles[address(0xae7ab96520DE3A18E5e111B5EaAb095312D7fE84)] = AggregatorV3Interface(address(0xCfE54B5cD566aB89272946F602D76Ea879CAb4a8));
    }


    modifier onlyEOA {
        require(msg.sender == tx.origin, "EOA");
        _;
    }

    function isSupportAsset(address asset) internal view returns (bool) {
        uint256 count = supportAssets.length;
        for(uint256 i = 0; i < count; i++) {
            if (supportAssets[i] == asset) {
                return true;
            }
        }
        return false;
    }

    function updateRouter(address _router) external onlyOwner {
        router = ICindexSwap(_router);
    }

    /*
     *@dev deposit assets
     */
    function deposit(DepositParams memory params) external onlyEOA nonReentrant returns (uint256){
        address tokenIn = params.tokenIn;
        uint256 amountIn = params.amountIn;
        string memory referralCode = params.referralCode;
        require(isSupportAsset(tokenIn), 'UnsupportedAsset');
        require(amountIn > 0, 'AmountInZero');
        ICindexSwap.SwapData[] memory swapData = params.swapData;
        require(swapData.length > 0, 'SwapDataZero');
        uint256 _sharePrePrice = sharePrePrice();
        TransferHelper.safeTransferFrom(tokenIn, msg.sender, address(this), amountIn);
        calProtocolFee();
        uint256 amount0 = amountIn * weights[0] / 100;
        uint256 amount1 = amountIn * weights[1] / 100;
        (uint256[] memory prices, uint8[] memory decimals) = getPrices();
        uint256[] memory newAmounts = _deposit(tokenIn, amount0, amount1, swapData);
        uint256 share = Formula.dot(newAmounts, prices, decimals) * _sharePrePrice / PRECISION;
        _mint(msg.sender, share);
        updateAssetAmounts();
        emit Deposit(msg.sender, share, tokenIn, amountIn, newAmounts[0], newAmounts[1], referralCode);
        return share;
    }

    /*
    *@dev deposit sDai and mint token
    */
    function depositUnderlying(uint256 amount0) external onlyEOA nonReentrant returns (uint256){
        require(amount0 > 0, 'Amount0InZero');
        uint256 _sharePrePrice = sharePrePrice();
        calProtocolFee();
        (uint256[] memory prices, uint8[] memory decimals) = getPrices();
        require(weights[0] > 0 && weights[1] > 0, 'WeightZero');
        uint256 amount1 = amount0 * prices[0] * weights[1] * (10 ** decimals[1]) / (weights[0] * prices[1] * (10 ** decimals[0]));
        uint256[] memory amounts = new uint256[](2);
        amounts[0] = amount0;
        amounts[1] = amount1;
        TransferHelper.safeTransferFrom(SavingsDaiMarket.sDAI, msg.sender, address(this), amounts[0]);
        TransferHelper.safeTransferFrom(StEthMarket.stETH, msg.sender, address(this), amounts[1]);
        uint256 share = Formula.dot(amounts, prices, decimals) * _sharePrePrice;
        _mint(msg.sender, share);
        updateAssetAmounts();
        emit DepositUnderlying(msg.sender, share, amounts[0], amounts[1]);
        return share;
    }

    /*
     *@dev withdraw shares
     */
    function withdraw(WithdrawParams memory params) external onlyEOA nonReentrant {
        uint256 share = params.share;
        address tokenOut = params.tokenOut;
        require(isSupportAsset(tokenOut), 'UnsupportedAsset');
        require(share > 0, 'ShareZero');
        ICindexSwap.SwapData[] memory swapData = params.swapData;
        require(swapData.length > 0, 'SwapDataZero');
        uint256 totalSupply = totalSupply();
        calProtocolFee();
        //The number of coins to be withdrawn
        uint256[] memory amounts = getPoolAmounts();
        uint256 amount0 = amounts[0] * share / totalSupply;//sDAI
        uint256 amount1 = amounts[1] * share / totalSupply;//stETH
        _burn(msg.sender, share);
        //First change sDAI to DAI, then swap to the target currency
        ICindexSwap.SwapData memory stEthSwapData;
        if (tokenOut == DAI) {
            SavingsDaiMarket.redeem(amount0, msg.sender, address(this));
            stEthSwapData = swapData[0];
        } else {
            uint256 amount = SavingsDaiMarket.redeem(amount0, address(router), address(this));
            router.swap(DAI, amount, swapData[0]);
            stEthSwapData = swapData[1];
        }
        //Then swap stETH directly to the target currency
        TransferHelper.safeTransfer(STETH, address(router), amount1);
        router.swap(STETH, amount1, stEthSwapData);
        updateAssetAmounts();
        emit Withdraw(msg.sender, share, tokenOut, share, amount0, amount1);
    }

    /*
    *@dev withdraw share get underlying assets
    */
    function withdrawUnderlying(uint256 share) external onlyEOA nonReentrant {
        require(share > 0, 'ShareZero');
        calProtocolFee();
        uint256 totalSupply = totalSupply();
        uint256[] memory amounts = getPoolAmounts();
        uint256 amount0Out = amounts[0] * share / totalSupply;
        uint256 amount1Out = amounts[1] * share / totalSupply;
        _burn(msg.sender, share);
        address token0Out = underlyingTokens[0];
        address token1Out = underlyingTokens[1];
        TransferHelper.safeTransfer(token0Out, msg.sender, amount0Out);
        TransferHelper.safeTransfer(token1Out, msg.sender, amount1Out);
        updateAssetAmounts();
        emit WithdrawUnderlying(msg.sender, share, amount0Out, amount1Out);
    }

    function updateAssetAmounts() internal {
        uint256 sDaiAmount = SavingsDaiMarket.balanceOf(address(this));
        uint256 exchangeRate = SavingsDaiMarket.exchangeRate();
        uint256 sETHAmount = StEthMarket.balanceOf(address(this));
        assetAmounts[SavingsDaiMarket.sDAI] = sDaiAmount * exchangeRate;
        assetAmounts[STETH] = sETHAmount;
    }

    /*
     *@dev Calculate fees
     */
    function calProtocolFee() internal {
        if (totalSupply() > 0) {
            //sDAI amount
            uint256 sDaiAmount = SavingsDaiMarket.balanceOf(address(this));
            uint256 exchangeRate = SavingsDaiMarket.exchangeRate();
            //stETH amount
            uint256 sETHAmount = StEthMarket.balanceOf(address(this));
            uint256 amount0 = assetAmounts[SavingsDaiMarket.sDAI];
            uint256 amount1 = assetAmounts[StEthMarket.stETH];
            //Interest-earning assets during this period
            uint256 sDaiFeeAmount = 0;
            uint256 sETHFeeAmount = 0;
            if (sDaiAmount * exchangeRate > amount0) {
                sDaiFeeAmount = (sDaiAmount * exchangeRate - amount0) * protocolFee / 100 / exchangeRate;
                TransferHelper.safeTransfer(SavingsDaiMarket.sDAI, PROTOCOL_FEE_RESERVE, sDaiFeeAmount);
            }
            if (sETHAmount > amount1) {
                sETHFeeAmount = (sETHAmount - amount1) * protocolFee / 100;
                TransferHelper.safeTransfer(STETH, PROTOCOL_FEE_RESERVE, sETHFeeAmount);

            }
            emit ProtocolFee(sDaiFeeAmount, sETHFeeAmount);
        }
    }
    
    function _deposit(address tokenIn, uint256 amount0, uint256 amount1, ICindexSwap.SwapData[] memory swapData) internal returns(uint256[] memory) {
        uint256[] memory amounts = new uint256[](2);
        uint256 sDaiAmount = _depositSavingDai(tokenIn, amount0, swapData);
        uint256 stEthAmount = _depositStEth(tokenIn, amount1, swapData);
        amounts[0] = sDaiAmount;
        amounts[1] = stEthAmount;
        return amounts;
    }

    /*
     *@dev Calculate the share corresponding to each price
     */
    function sharePrePrice() public view returns(uint256) {
        uint256[] memory poolAmounts = getPoolAmounts();
        (uint256[] memory prices, uint8[] memory decimals) = getPrices();
        uint256 value = Formula.dot(poolAmounts, prices, decimals);
        return totalSupply() > 0 && value > 0 ? totalSupply() * PRECISION / value : PRECISION;
    }

    /*
     *@dev query Pool Asset Amounts
     */
    function getPoolAmounts() public view returns (uint256[] memory) {
        uint256[] memory amounts = new uint256[](2);
        amounts[0] = IERC20(SavingsDaiMarket.sDAI).balanceOf(address(this));
        amounts[1] = IERC20(STETH).balanceOf(address(this));
        return amounts;
    }

    function _depositSavingDai(address token, uint256 amount, ICindexSwap.SwapData[] memory swapdata) internal returns (uint256) {
        if (amount == 0) {
            return 0;
        }
        if (token == DAI) {
            //The deposit is DAI, and the amount of coins in amount 0 does not require swap.
            return SavingsDaiMarket.deposit(amount, address(this));
        } else {
            uint256 beforeAmount = IERC20(DAI).balanceOf(address(this));
            // If it is not DAI, you need to perform two swaps. Change part of the tokenIn to DAI and store it in sDAI. Then change the other part of tokenIn to ETH and store it in stETH, which means you need to perform two swap operations.
            TransferHelper.safeTransfer(token, address(router), amount);
            ICindexSwap.SwapData memory _swapdata = swapdata[0];
            router.swap(token, amount, _swapdata);
            uint256 afterAmount = IERC20(DAI).balanceOf(address(this));

            return SavingsDaiMarket.deposit(afterAmount - beforeAmount, address(this));
        }
    }

    function _depositStEth(address token, uint256 amount, ICindexSwap.SwapData[] memory swapdata) internal returns (uint256) {
        if (amount == 0) {
            return 0;
        }
        uint256 beforeAmount = address(this).balance;
        TransferHelper.safeTransfer(token, address(router), amount);
        ICindexSwap.SwapData memory _swapdata;
        if (token == DAI) {
            _swapdata = swapdata[0];
        } else {
            _swapdata = swapdata[1];
        }
        router.swap(token, amount, _swapdata);
        uint256 afterAmount = address(this).balance;
        return StEthMarket.submit(address(0), afterAmount - beforeAmount);
    }

    function getPrices() public view returns (uint256[] memory, uint8[] memory) {
        uint256[] memory prices = new uint256[](2);
        uint8[] memory decimals = new uint8[](2);
        // query dai oracle price
        (uint256 price0, uint8 decimals0) = _getPrice(oracles[DAI]);
        // query dai converter sDai exchange rate
        uint256 exchangeRate = SavingsDaiMarket.exchangeRate();
        prices[0] = price0 * exchangeRate / PRECISION;
        decimals[0] = decimals0;
        (uint256 price1, uint8 decimals1) = _getPrice(oracles[STETH]);
        prices[1] = price1;
        decimals[1] = decimals1;

        return (prices, decimals);
    }

    function _getPrice(AggregatorV3Interface feed) internal view returns (uint256, uint8) {
        (uint80 roundId,int256 price,,uint256 updatedAt, uint80 answeredInRound) = feed.latestRoundData();
        require(price > 0, "Chainlink price <= 0");
        require(updatedAt != 0, "Incomplete round");
        require(answeredInRound >= roundId, "Stale price");
        uint8 decimals = feed.decimals();
        return (uint256(price), decimals);
    }

    function getProtocolFeeReserve() public view returns (address) {
        return PROTOCOL_FEE_RESERVE;
    }

    receive() external payable {

    }

    fallback() external payable{

    }
}

// File: contracts/exchange/ICindexSwap.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity ^0.8.9;

interface ICindexSwap {

    struct SwapData {
        address extRouter;
        bytes extCalldata;
        bool needScale;
    }
    function swap(address tokenIn, uint256 amountIn, SwapData calldata swapData) external payable;
}
// File: contracts/markets/StEthMarket.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity ^0.8.9;

interface Lido {
    /**
     * @notice Send funds to the pool with optional _referral parameter
     * @dev This function is alternative way to submit funds. Supports optional referral address.
     * @return Amount of StETH shares generated
     */
    function submit(address) external payable returns (uint256);

    /**
     * @return the amount of tokens owned by the `_account`.
     *
     * @dev Balances are dynamic and equal the `_account`'s share in the amount of the
     * total Ether controlled by the protocol. See `sharesOf`.
     */
    function balanceOf(address) external view returns (uint256);

    function getPooledEthByShares(uint256) external view returns (uint256);
}

library StEthMarket {

    address constant stETH = 0xae7ab96520DE3A18E5e111B5EaAb095312D7fE84;

    function balanceOf(address _account) internal view returns (uint256) {
        return Lido(stETH).balanceOf(_account);     
    }    

    function submit(address _referral, uint256 _value) internal returns (uint256) {
        uint256 shareAmount = Lido(stETH).submit{ value: _value }(_referral);
        return Lido(stETH).getPooledEthByShares(shareAmount);
    }
}
// File: contracts/markets/SavingsDaiMarket.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity ^0.8.9;
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

interface ISDai {
    function convertToAssets(uint256) external view returns (uint256);
    function convertToShares(uint256) external view returns (uint256);
    function deposit(uint256,address) external returns (uint256);
    function redeem(uint256,address,address) external returns (uint256);
    function balanceOf(address) external view returns (uint256);
}

library SavingsDaiMarket {

    address constant sDAI = 0x83F20F44975D03b1b09e64809B757c47f942BEeA;

    address constant DAI = 0x6B175474E89094C44Da98b954EedeAC495271d0F;

    function exchangeRate() internal view returns (uint256) {
        return ISDai(sDAI).convertToAssets(1e18);
    }

    function balanceOf(address account) internal view returns (uint256) {
        return ISDai(sDAI).balanceOf(account);
    }

    function deposit(uint256 amount, address receiver) internal returns(uint256) {
        _approve(amount);
        return ISDai(sDAI).deposit(amount, receiver);
    }

    function redeem(uint256 amount, address receiver, address sender) internal returns(uint256) {
        return ISDai(sDAI).redeem(amount, receiver, sender);
    }

    function _approve(uint256 _amount) internal {
        if (IERC20(DAI).allowance(address(this), sDAI) < _amount){
            IERC20(DAI).approve(sDAI, ~uint256(0));
        }
    }
}
// File: contracts/Formula.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.9;


library Formula {

    /*
     *@dev calculate the number of missing tokens
     *@param currentRatios current proportion
     *@param amounts current number of tokens
     *@param targetRatio target proportion
     */
    function calDeltaAmounts(uint256[] memory currentRatios, uint256[] memory targetRatios, uint256[] memory amounts) internal pure returns(uint256[] memory) {
        uint256 num = currentRatios.length;
        int256 minDeltaRatio = 1e18;
        uint256 minIndex = 0;
        for(uint256 i = 0; i < num; i++) {
            uint256 tmp = targetRatios[i] * currentRatios[0] / targetRatios[0];
            int256 deltaRatio = tmp > currentRatios[i] ? int256((tmp - currentRatios[i]) * 1e18 / currentRatios[i]) : -int256((currentRatios[i] - tmp) * 1e18 / currentRatios[i]);
            if (deltaRatio < minDeltaRatio) {
                minDeltaRatio = deltaRatio;
                minIndex = i;
            }
        }
        uint256[] memory deltaAmts = new uint256[](num);
        for(uint256 i = 0; i < num; i++) {
            uint256 amount = amounts[minIndex] * targetRatios[i] / targetRatios[minIndex];
            deltaAmts[i] = amount > amounts[i] ? amount - amounts[i] : 0;
        }
        return deltaAmts;
    }

    /*
     *@dev calculate the number of missing tokens and the maximum missing token index
     *@param currentRatios current proportion
     *@param targetRatio target proportion
     *@param amounts current number of tokens
     *@param prices token prices
     *@param decimals  token precision
     */
    function calMaxGapCoin(uint256[] memory currentRatios, uint256[] memory targetRatios, uint256[] memory amounts, uint256[] memory prices, uint8[] memory decimals) internal pure returns(uint256[] memory, uint256) {
        uint256 num = currentRatios.length;
        int256 minDeltaRatio = 1e18;
        uint256 minIndex = 0;
        for(uint256 i = 0; i < num; i++) {
            uint256 tmp = targetRatios[i] * currentRatios[0] / targetRatios[0];
            int256 deltaRatio = tmp > currentRatios[i] ? int256((tmp - currentRatios[i]) * 1e18 / currentRatios[i]) : -int256((currentRatios[i] - tmp) * 1e18 / currentRatios[i]);
            if (deltaRatio < minDeltaRatio) {
                minDeltaRatio = deltaRatio;
                minIndex = i;
            }
        }
        uint256 maxIndex = 0;
        uint256[] memory deltaAmts = new uint256[](num);
        uint256 maxValue = 0;
        for(uint256 i = 0; i < num; i++) {
            uint256 amount = amounts[minIndex] * targetRatios[i] / targetRatios[minIndex];
            deltaAmts[i] = amount > amounts[i] ? amount - amounts[i] : 0;
            uint256 deltaValue = deltaAmts[i] * prices[i] / decimals[i];
            if (deltaValue > maxValue) {
                maxValue = deltaValue;
                maxIndex = i;
            }
        }
        return (deltaAmts, maxIndex);
    }

    /*
     *@dev calculate token ratio based on user added tokens
     *
     *@param amounts current number of tokens
     *@param targetRatio target proportion
     *@param deltaAmt  missing tokens
     *@param index maximum missing tokens index
     *@param value user adding Tokens
     */
    function calIncrease(uint256[] memory amounts, uint256[] memory targetRatios, uint256[] memory deltaAmt, uint256 index, uint256 value) internal pure returns (uint256[] memory) {
        uint256 num = amounts.length;
        uint256[] memory t = new uint256[](num);
        t[index] = value;
        if (t[index] > deltaAmt[index]) {
            for (uint256 i = 0; i < num; i++) {
                t[i] = deltaAmt[i] + targetRatios[i] * (t[index] - deltaAmt[index]) / targetRatios[index];
            }
        } else {
            for (uint256 i = 0; i < num; i++) {
                t[i] = deltaAmt[i] * t[index] / deltaAmt[index];
            }
        }
        
        return t;
    }

    /*
     *@dev calculate the value of the pool
     *
     *@param amounts current number of tokens
     *@param targetRatio target proportion
     *@param prices token prices
     *@param decimals  token precision
     */
    function dot(uint256[] memory amounts, uint256[] memory prices, uint8[] memory decimals) internal pure returns (uint256) {
        uint256 result = 0;
        for (uint256 i = 0; i < amounts.length; i++) {
            result += amounts[i] * prices[i] / (10 ** decimals[i]);
        }
        return result;
    }

    /*
     *@dev calculate token constant
     *
     *@param decreaseTokenRatio The proportion of tokens that need to be reduced
     *@param prices token prices
     *@param decimals  token Precision
     *@param lastValue remaining pool value
     *@param keepAmounts unchanged token amounts
     */
    function calMultiple(uint256[] memory ratios, uint256[] memory prices, uint8[] memory decimals, uint256 value, uint256[] memory amounts) internal pure returns (uint256) {
        require(ratios.length == prices.length && ratios.length == amounts.length, "Invalid input length");
        
        uint256 dotProductKP = 0;
        uint256 dotProductTP = 0;
        
        for (uint256 i = 0; i < ratios.length; i++) {
            dotProductKP += ratios[i] * prices[i] * 1e18 / (10 ** decimals[i]);
            dotProductTP += amounts[i] * prices[i] / (10 ** decimals[i]);
        }
        
        return (value - dotProductTP) * 1e36 / dotProductKP;
    }

    /*
     *@dev calculate the proportion of reduced tokens
     *
     *@param amounts current number of tokens
     *@param targetRatio target proportion
     *@param prices token prices
     *@param decimals  token precision
     *@param lastValue remaining pool value
     */
    function calDecrease(uint256[] memory amounts, uint256[] memory targetRatios, uint256[] memory prices, uint8[] memory decimals, uint256 value) internal pure returns (uint256[] memory) {
        uint256[] memory Kt = targetRatios;
        uint256 num = amounts.length;
        uint256[] memory T1 = new uint256[](num);
        uint256[] memory res = new uint256[](num);
        while (true) {
            uint256 mul_t = calMultiple(Kt, prices, decimals, value, T1);
            uint256[] memory Tt = new uint256[](num);
            int256[] memory F1 = new int256[](num);
            bool isReduce = false;
            for (uint256 j = 0; j < num; j++) {
                Tt[j] = Kt[j] > 0 ? mul_t * Kt[j] / 1e18 : amounts[j];
                if (Tt[j] > amounts[j]) {
                    isReduce = true;
                }
                if (Tt[j] < amounts[j]) {
                    F1[j] = 1;
                } else {
                    T1[j] = amounts[j];
                }
            }
            if (isReduce) {
                for (uint256 j = 0; j < num; j++) {
                    Kt[j] = F1[j] > 0 ? targetRatios[j] : 0;
                }
            } else {
                for (uint256 j = 0; j < num; j++) {
                    Tt[j] = Kt[j] > 0 ? mul_t * Kt[j] / 1e18 : amounts[j];
                    res[j] = amounts[j] - Tt[j];
                }
                break;
            }
        }
        return res;
    }
}
// File: contracts/TransferHelper.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity ^0.8.9;
import "@openzeppelin/contracts/interfaces/IERC20.sol";

library TransferHelper {

    address internal constant NATIVE = address(0);
    uint256 internal constant LOWER_BOUND_APPROVAL = type(uint96).max / 2;

    /// @notice Transfers tokens from the targeted address to the given destination
    /// @notice Errors with 'STF' if transfer fails
    /// @param token The contract address of the token to be transferred
    /// @param from The originating address from which the tokens will be transferred
    /// @param to The destination address of the transfer
    /// @param value The amount to be transferred
    function safeTransferFrom(
        address token,
        address from,
        address to,
        uint256 value
    ) internal {
        (bool success, bytes memory data) =
        token.call(abi.encodeWithSelector(IERC20.transferFrom.selector, from, to, value));
        require(success && (data.length == 0 || abi.decode(data, (bool))), 'STF');
    }

    /// @notice Transfers tokens from msg.sender to a recipient
    /// @dev Errors with ST if transfer fails
    /// @param token The contract address of the token which will be transferred
    /// @param to The recipient of the transfer
    /// @param value The value of the transfer
    function safeTransfer(
        address token,
        address to,
        uint256 value
    ) internal {
        (bool success, bytes memory data) = token.call(abi.encodeWithSelector(IERC20.transfer.selector, to, value));
        require(success && (data.length == 0 || abi.decode(data, (bool))), 'ST');
    }

    /// @notice Approves the stipulated contract to spend the given allowance in the given token
    /// @dev Errors with 'SA' if transfer fails
    /// @param token The contract address of the token to be approved
    /// @param to The target of the approval
    /// @param value The amount of the given token the target will be allowed to spend
    function safeApprove(
        address token,
        address to,
        uint256 value
    ) internal {
        (bool success, bytes memory data) = token.call(abi.encodeWithSelector(IERC20.approve.selector, to, value));
        require(success && (data.length == 0 || abi.decode(data, (bool))), 'SA');
    }

    /// @notice Transfers ETH to the recipient address
    /// @dev Fails with `STE`
    /// @param to The destination of the transfer
    /// @param value The value to be transferred
    function safeTransferETH(address to, uint256 value) internal {
        (bool success, ) = to.call{value: value}(new bytes(0));
        require(success, 'STE');
    }

    function safeApproveInf(address token, address to) internal {
        if (token == NATIVE) return;
        if (IERC20(token).allowance(address(this), to) < LOWER_BOUND_APPROVAL) {
            safeApprove(token, to, 0);
            safeApprove(token, to, type(uint256).max);
        }
    }
}


// File: @chainlink/contracts/src/v0.8/interfaces/AggregatorV3Interface.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

interface AggregatorV3Interface {
  function decimals() external view returns (uint8);

  function description() external view returns (string memory);

  function version() external view returns (uint256);

  function getRoundData(
    uint80 _roundId
  ) external view returns (uint80 roundId, int256 answer, uint256 startedAt, uint256 updatedAt, uint80 answeredInRound);

  function latestRoundData()
    external
    view
    returns (uint80 roundId, int256 answer, uint256 startedAt, uint256 updatedAt, uint80 answeredInRound);
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == _ENTERED;
    }
}

// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.openzeppelin.com/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * The default value of {decimals} is 18. To change this, you should override
 * this function so it returns a different value.
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the default value returned by this function, unless
     * it's overridden.
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(address from, address to, uint256 amount) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `from` to `to`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(address from, address to, uint256 amount) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(from, to, amount);

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
            // Overflow not possible: the sum of all balances is capped by totalSupply, and the sum is preserved by
            // decrementing then incrementing.
            _balances[to] += amount;
        }

        emit Transfer(from, to, amount);

        _afterTokenTransfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        unchecked {
            // Overflow not possible: balance + amount is at most totalSupply + amount, which is checked above.
            _balances[account] += amount;
        }
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
            // Overflow not possible: amount <= accountBalance <= totalSupply.
            _totalSupply -= amount;
        }

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(address owner, address spender, uint256 amount) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(address owner, address spender, uint256 amount) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(address from, address to, uint256 amount) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(address from, address to, uint256 amount) internal virtual {}
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/interfaces/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (interfaces/IERC20.sol)

pragma solidity ^0.8.0;

import "../token/ERC20/IERC20.sol";

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}